
11 Linear Constrained Optimization

Linear programming involves solving optimization problems with linear objec-
tive functions and linear constraints. Many problems are naturally described by
linear programs, including problems from fields as diverse as transportation,
communication networks, manufacturing, economics, and operations research.
Many problems that are not naturally linear can often be approximated by linear
programs. Several methods have been developed for exploiting the linear struc-
ture. Modern techniques and hardware can globally minimize problems with
millions of variables and millions of constraints.1

1 This chapter is a short intro-
duction to linear programs and
one variation of the simplex al-
gorithm used to solve them. Sev-
eral textbooks are dedicated en-
tirely to linear programs, includ-
ing R. J. Vanderbei, Linear Program-
ming: Foundations and Extensions,
4th ed. Springer, 2014. There are a
variety of packages for solving lin-
ear programs, such as Convex.jl
and JuMP.jl, both of which in-
clude interfaces to open-source
and commercial solvers.

11.1 Problem Formulation

A linear programming problem, called a linear program,2 can be expressed in 2 A quadratic program is a gen-
eralization of a linear program,
where the objective function is
quadratic and the constraints are
linear. Common approaches for
solving such problems include
some of the algorithms discussed
in earlier chapters, including the
interior point method, augmented
Lagrange method, and conjugate
gradient. The simplexmethod, cov-
ered in this chapter, has also been
adapted for optimizing quadratic
programs. J. Nocedal and S. J.
Wright, Numerical Optimization,
2nd ed. Springer, 2006.

several forms. Each linear program consists of a linear objective function and a
set of linear constraints:

minimize
x

c⊤x

subject to w
(i)⊤
LE x ≤ bi for i ∈ {1, 2, . . .}

w
(j)⊤
GE x ≥ bj for j ∈ {1, 2, . . .}

w
(k)⊤
EQ x = bk for k ∈ {1, 2, . . .}

(11.1)

where i, j, and k vary over finite sets of constraints. Such an optimization problem
is given in example 11.1. Transforming real problems into this mathematical form
is often nontrivial. This text focuses on the algorithms for obtaining solutions, but
other texts discuss how to go about modeling real problems.3 Several interesting 3 See, for example, H. P. Williams,

Model Building in Mathematical Pro-
gramming, 5th ed. Wiley, 2013.conversions are given in example 11.2.



190 chapter 11. l inear constrained optimization

The following problem has a linear objective and linear constraints, making
it a linear program.

minimize
x1,x2,x3

2x1 − 3x2 + 7x3

subject to 2x1 + 3x2 − 8x3 ≤ 5

4x1 + x2 + 3x3 ≤ 9

x1 − 5x2 − 3x3 ≥ −4

x1 + x2 + 2x3 = 1

Example 11.1. An example linear
program.

Many problems can be converted into linear programs that have the same
solution. Two examples are L1 and L∞ minimization problems:

minimize‖Ax− b‖1 minimize‖Ax− b‖∞

The first problem is equivalent to solving

minimize
x,s

1⊤s

subject to Ax− b ≤ s

Ax− b ≥ −s

with the additional variables s.
The second problem is equivalent to solving

minimize
x,t

t

subject to Ax− b ≤ t1

Ax− b ≥ −t1

with the additional variable t.

Example 11.2. Common normmin-
imization problems that can be con-
verted into linear programs.

11.1.1 General Form
We can write linear programs more compactly using matrices and arrive at the
general form:4 4 Here, each constraint is element-

wise. For example, in writing
a ≤ b,

we mean ai ≤ bi for all i.

minimize
x

c⊤x

subject to ALEx ≤ bLE
AGEx ≥ bGE
AEQx = bEQ

(11.2)

11.1.2 Standard Form
The general linear program given in equation (11.2) can be converted into standard
form where all constraints are less-than inequalities and the design variables are
nonnegative

minimize
x

c⊤x

subject to Ax ≤ b

x ≥ 0

(11.3)

Greater-than inequalities are inverted, and equality constraints are split in two

AGEx ≥ bGE → −AGEx ≤ −bGE

AEQx = bEQ →







AEQx ≤ bEQ
−AEQx ≤ −bEQ

(11.4)

We must ensure that all x entries are nonnegative as well. Suppose we start
with a linear program where x is not constrained to be nonnegative:

minimize
x

c⊤x

subject to Ax ≤ b
(11.5)

We replace x with x+ − x− and constrain x+ ≥ 0 and x− ≥ 0:

minimize
x+ ,x−

[

c⊤ −c⊤
]
[

x+

x−

]

subject to
[

A −A
]
[

x+

x−

]

≤ b

[

x+

x−

]

≥ 0

(11.6)

The linear objective function c⊤x forms a flat ramp. The function increases in
the direction of c, and, as a result, all contour lines are perpendicular to c and
parallel to one another as shown in figure 11.1.

c

c ⊤
x = const

Figure 11.1. The contours of a lin-
ear objective function c⊤x, which
increase in the direction of c.

A single inequality constraint w⊤x ≤ b forms a half-space, or a region on one
side of a hyperplane. The hyperplane is perpendicular to w and is defined by
w⊤x = b as shown in figure 11.2. The region w⊤x > b is on the +w side of the
hyperplane, whereas w⊤x < b is on the −w side of the hyperplane.

w

w⊤x < b

w⊤x > b

w ⊤
x =

b

Figure 11.2. A linear constraint.

X

Figure 11.3. The intersection of lin-
ear constraints is a convex set.

Half-spaces are convex sets (see appendix C.3), and the intersection of convex
sets is convex, as shown in figure 11.3. Thus, the feasible set of a linear program
will always form a convex set. Convexity of the feasible set, along with convexity
of the objective function, implies that if we find a local feasible minimum, it is
also a global feasible minimum.

The feasible set is a convex region enclosed by flat faces. Depending on the
region’s configuration, the solution can lie at a vertex, on an edge, or on an entire
face. If the problem is not properly constrained, the solution can be unbounded,

and, if the system is over-constrained, there is no feasible solution. Several such
cases are shown in figure 11.4.

x∗

−c

x1

x
2

One Solution

−c

x1

Unbounded Solution

x∗

−c

x1

Infinite Solutions

−c

x1

No Solution

Figure 11.4. Several different lin-
ear problem forms with different
solutions.

11.1.3 Equality Form
Linear programs are often solved in a third form, the equality form

minimize
x

c⊤x

subject to Ax = b

x ≥ 0

(11.7)

where x and c each have n components, A is an m × n matrix, and b has m

components. In other words, we have n nonnegative design variables and a
system of m equations defining equality constraints.

The equality form has constraints in two parts. The first, Ax = b, forces the
solution to lie in an affine subspace.5 Such a constraint is convenient because 5 Informally, an affine subspace is a

vector space that has been trans-
lated such that its origin in a higher-
dimensional space is not necessar-
ily 0.

search techniques can constrain themselves to the constrained affine subspace to
remain feasible. The second part of the constraints requires x ≥ 0, which forces
the solution to lie in the positive quadrant. The feasible set is thus the nonnegative
portion of an affine subspace. Example 11.3 provides a visualization for a simple
linear program.

Any linear program in standard form can be transformed to equality form. The
constraints are converted to:

Ax ≤ b → Ax + s = b, s ≥ 0 (11.8)

Consider the standard-form linear program:

minimize
x

x

subject to x ≥ 1

When we convert this to equality form, we get

minimize
x,s

x

subject to x− s = 1

x, s ≥ 0

The equality constraint requires that feasible points fall on the line x− s =

1. That line is a one-dimensional affine subspace of the two-dimensional
Euclidean space.

Example 11.3. Feasible sets for
the equality form are hyperplanes.

2 4

2

x

s

by introducing slack variables s. These variables take up the extra slack to enforce
equality.

Starting with a linear program:

minimize
x

c⊤x

subject to Ax ≤ b

x ≥ 0

(11.9)

We introduce the slack variables:

minimize
x,s

[

c⊤ 0⊤
]
[

x

s

]

subject to
[

A I
]
[

x

s

]

= b

[

x

s

]

≥ 0

(11.10)

Example 11.4 demonstrates converting from standard to equality form.

Consider the linear program

minimize
x

5x1 + 4x2

subject to 2x1 + 3x2 ≤ 5

4x1 + x2 ≤ 11

To convert to equality form, we first introduce two slack variables:

minimize
x,s

5x1 + 4x2

subject to 2x1 + 3x2 + s1 = 5

4x1 + x2 + s2 = 11

s1, s2 ≥ 0

We then split x:

minimize
x+ ,x− ,s

5(x+1 − x−1 ) + 4(x+2 − x−2 )

subject to 2(x+1 − x−1 ) + 3(x+2 − x−2 ) + s1 = 5

4(x+1 − x−1 ) + (x+2 − x−2 ) + s2 = 11

x+1 , x−1 , x+2 , x−2 , s1, s2 ≥ 0

Example 11.4. Converting a linear
program to equality form.

11.2 Simplex Algorithm

The simplex algorithm solves linear programs by moving from vertex to vertex of
the feasible set.6 The method is guaranteed to arrive at an optimal solution so

6 The simplex algorithm was orig-
inally developed in the 1940s by
George Dantzig. A history of the
development can be found here:
G. B. Dantzig, “Origins of the Sim-
plex Method,” in A History of Sci-
entific Computing, S. G. Nash, ed.,
ACM, 1990, pp. 141–151.

long as the linear program is feasible and bounded.
The simplex algorithm operates on equality-form linear programs (Ax = b,

x ≥ 0). We assume that the rows of A are linearly independent.7 We also assume
7 A matrix whose rows are linearly
independent is said to have full
row rank. Linear independence is
achieved by removing redundant
equality constraints.

that the problem has no more equality constraints than it has design variables
(m ≤ n), which ensures that the problem is not over constrained. A preprocessing
phase guarantees that A satisfies these conditions.

11.2.1 Vertices
Linear programs in equality form have feasible sets in the form of convex polytopes,
which are geometric objects with flat faces. These polytopes are formed by the
intersection of the equality constraints with the positive quadrant. Associated
with a polytope are vertices, which are points in the feasible set that do not lie
between any other points in the feasible set.

The feasible set consists of several different types of design points. Points on
the interior are never optimal because they can be improved by moving along
−c. Points on faces can be optimal only if the face is perpendicular to c. Points
on faces not perpendicular to c can be improved by sliding along the face in the
direction of the projection of −c onto the face. Similarly, points on edges can be
optimal only if the edge is perpendicular to c, and can otherwise be improved
by sliding along the projection of −c onto the edge. Finally, vertices can also be
optimal.

The simplex algorithm produces an optimal vertex. If a linear program has
a bounded solution, then it also contains at least one vertex. Furthermore, at
least one solution must lie at a vertex. In the case where an entire edge or face is
optimal, a vertex-solution is just as good as any other.

X

vertex

Figure 11.5. In twodimensions, any
vertex will have at least two active
constraints.

Every vertex for a linear program in equality form can be uniquely defined by
n−m components of x that equal zero. These components are actively constrained
by xi ≥ 0. Figure 11.5 visualizes the active constraints needed to identify a vertex.

The equality constraint Ax = b has a unique solution when A is square.
We have assumed that m ≤ n, so choosing m design variables and setting the
remaining variables to zero effectively removes n−m columns of A, yielding an
m×m constraint matrix (see example 11.5).

For a problem with 5 design variables and 3 constraints, setting 2 variables
to zero uniquely defines a point.






a11 a12 a13 a14 a15

a21 a22 a23 a24 a25

a31 a32 a33 a34 a35















x1

0

x3

x4

0










=






a11 a13 a14

a21 a23 a24

a31 a33 a34











x1

x3

x4




 =






b1

b2

b3






Example 11.5. Setting n−m com-
ponents of x to zero can uniquely
define a point.

The indices into the components {1, . . . , n} of any vertex can be partitioned
into two sets, B and V , such that:

• The design values associated with indices in V are zero:

i ∈ V =⇒ xi = 0 (11.11)

• The design values associated with indices in B may or may not be zero:

i ∈ B =⇒ xi ≥ 0 (11.12)

• B has exactly m elements and V has exactly n−m elements.

We use xB to refer to the vector consisting of the components of x that are in B
and xV to refer to the vector consisting of the components of x that are in V . Note
that xV = 0.

The vertex associated with a partition (B,V) can be obtained using the m×m

matrix AB formed by the m columns of A selected by B:8 8 If B and V identify a vertex, then
the columns of AB must be linearly
independent because Ax = b must
have exactly one solution. Hence,
ABxB = b must have exactly one
solution. This linear independence
guarantees that AB is invertible.

Ax = ABxB = b → xB = A−1
B b (11.13)

Knowing xB is sufficient to construct x; the remaining design variables are zero. Al-
gorithm 11.1 implements this procedure, and example 11.6 demonstrates verifying
that a given design point is a vertex.

mutable struct LinearProgram
A
b
c

end
function get_vertex(B, LP)

A, b, c = LP.A, LP.b, LP.c
b_inds = sort!(collect(B))
AB = A[:,b_inds]
xB = AB\b
x = zeros(length(c))
x[b_inds] = xB
return x

end

Algorithm 11.1. A method for ex-
tracting the vertex associated with
a partition B and a linear program
LP in equality form. We introduce
the special type LinearProgram for
linear programs.

While every vertex has an associated partition (B,V), not every partition corre-
sponds to a vertex. A partition corresponds to a vertex only if AB is nonsingular
and the design obtained by applying equation (11.13) is feasible.9 Identifying par- 9 For example, B = {1, 2} for the

constraints
[

1 2 0
1 2 1

]




x1

x2

x3



 =

[
1
1

]

corresponds to
[

1 2
1 2

] [
x1

x2

]

=

[
1
1

]

which does not produce an invert-
ible AB and does not have a unique
solution.

titions that correspond to vertices is nontrivial, and we show in section 11.2.4 that
finding such a partition involves solving a linear program! The simplex algorithm
operates in two phases—an initialization phase that identifies a vertex partition and
an optimization phase that transitions between vertex partitions toward a partition
corresponding to an optimal vertex. We will discuss both of these phases later in
this section.

Consider the constraints:





1 1 1 1

0 −1 2 3

2 1 2 −1




 x =






2

−1

3




 , x ≥ 0

Consider the design point x = [1, 1, 0, 0]. We can verify that x is feasible
and that it has nomore than three nonzero components. We can choose either
B = {1, 2, 3} or B = {1, 2, 4}. Both

A{1,2,3} =






1 1 1

0 −1 2

2 1 2






and

A{1,2,4} =






1 1 1

0 −1 3

2 1 −1






are invertible. Thus, x is a vertex of the feasible set polytope.

Example 11.6. Verifying that a de-
sign point is a vertex for constraints
in equality form.

11.2.2 First-Order Necessary Conditions
The first-order necessary conditions (FONCs) for optimality are used to determine
when a vertex is optimal and to inform how to transition to a more favorable
vertex. We construct a Lagrangian for the equality form of the linear program:10

10 Note that in x ≥ 0 the polarity
of the inequality must be inverted
by multiplying both sides by −1,
yielding the negative sign in front
of µ. The Lagrangian can be de-
fined with either positive or neg-
ative λ.

L(x,µ,λ) = c⊤x− µ⊤x− λ⊤(Ax− b) (11.14)

with the following FONCs:

1. feasibility: Ax = b, x ≥ 0

2. dual feasibility: µ ≥ 0

3. complementary slackness: µ⊙ x = 0

4. stationarity: A⊤λ+ µ = c

The FONCs are sufficient conditions for optimality for linear programs. Thus,
if µ and λ can be computed for a given vertex and all four FONC equations are
satisfied, then the vertex is optimal.

We can decompose the stationarity condition into B and V components:

A⊤λ+ µ = c →







A⊤Bλ+ µB = cB
A⊤V λ+ µV = cV

(11.15)

We can choose µB = 0 to satisfy complementary slackness. The value of λ can
be computed from B:11 11 We use A−⊤ to refer to the trans-

pose of the inverse of A:

A−⊤ =
(

A−1
)⊤

=
(

A⊤
)−1

A⊤Bλ+ µB
︸︷︷︸

=0

= cB

λ = A−⊤B cB

(11.16)

We can use this to obtain

A⊤V λ+ µV = cV
µV = cV −A⊤V λ

µV = cV −
(

A−1
B AV

)⊤
cB

(11.17)

Knowing µV allows us to assess the optimality of the vertices. If µV contains
negative components, then dual feasibility is not satisfied and the vertex is sub-
optimal.

11.2.3 Optimization Phase
The simplex algorithmmaintains a partition (B,V), which corresponds to a vertex
of the feasible set polytope. The partition can be updated by swapping indices
between B and V . Such a swap equates to moving from one vertex along an edge
of the feasible set polytope to another vertex. If the initial partition corresponds
to a vertex and the problem is bounded, the simplex algorithm is guaranteed to
converge to an optimum.

A transition x → x′ between vertices must satisfy Ax′ = b. Starting with a
partition defined by B, we choose an entering index q ∈ V that is to enter B using
one of the heuristics described near the end of this section. The new vertex x′

must satisfy:
Ax′ = ABx′B + A{q}x

′
q = ABxB = Ax = b (11.18)

One leaving index p ∈ B in x′B becomes zero during the transition and is replaced
by the column of A corresponding to index q. This action is referred to as pivoting.

We can solve for the new design point

x′B = xB −A−1
B A{q}x

′
q (11.19)

A particular leaving index p ∈ B becomes active when:
(
x′B
)

p
= 0 = (xB)p −

(

A−1
B A{q}

)

p
x′q (11.20)

and is thus obtained by increasing xq = 0 to x′q with:

x′q =
(xB)p

(

A−1
B A{q}

)

p

(11.21)

The leaving index is obtained using theminimum ratio test, which computes for
each potential leaving index and selects the one with minimum x′q. We then swap
p and q between B and V . The edge transition is implemented in algorithm 11.2.

function edge_transition(LP, B, q)
A, b, c = LP.A, LP.b, LP.c
n = size(A, 2)
b_inds = sort(B)
n_inds = sort!(setdiff(1:n, B))
AB = A[:,b_inds]
d, xB = AB\A[:,n_inds[q]], AB\b

p, xq′ = 0, Inf
for i in 1 : length(d)

if d[i] > 0
v = xB[i] / d[i]
if v < xq′

p, xq′ = i, v
end

end
end
return (p, xq′)

end

Algorithm 11.2. Amethod for com-
puting the index p and the new
coordinate value x′q obtained by
increasing index q of the vertex
defined by the partition B in the
equality-form linear program LP.

The effect that a transition has on the objective function can be computed using
x′q. The objective function value at the new vertex is12 12 Here, we used the fact that λ =

A−⊤B cB and that A⊤{q}λ = cq − µq.
c⊤x′ = c⊤B x′B + cqx′q (11.22)

= c⊤B
(

xB −A−1
B A{q}x

′
q

)

+ cqx′q (11.23)
= c⊤B xB − c⊤BA−1

B A{q}x
′
q + cqx′q (11.24)

= c⊤B xB −
(
cq − µq

)
x′q + cqx′q (11.25)

= c⊤x + µqx′q (11.26)

Choosing an entering index q decreases the objective function value by

c⊤x′ − c⊤x = µqx′q (11.27)

The objective function decreases only if µq is negative. In order to progress to-
ward optimality, we must choose an index q in V such that µq is negative. If all
components of µV are non-negative, we have found a global optimum.

Since there can be multiple negative entries in µV , different heuristics can be
used to select an entering index:13

13 Modern implementations use
more sophisticated rules. For ex-
ample, see J. J. Forrest and D. Gold-
farb, “Steepest-Edge Simplex Algo-
rithms for Linear Programming,”
Mathematical Programming, vol. 57,
no. 1, pp. 341–374, 1992.

• Greedy heuristic, which chooses a q that maximally reduces c⊤x.

• Dantzig’s rule, which chooses the q with the most negative entry in µ. This rule
is easy to calculate, but it does not guarantee the maximum reduction in c⊤x.
It is also sensitive to scaling of the constraints.14 14 For constraint A⊤x = b →

αA⊤x = αb, α > 0, we do
not change the solution but the
Lagrange multipliers are scaled,
λ→ α−1λ.

• Bland’s rule, which chooses the first q with a negative entry in µ. When used on
its own, Bland’s rule tends to result in poor performance in practice. However,
this rule can help us prevent cycles, which occur when we return to a vertex
we have visited before without decreasing the objective function. This rule is
usually used only after no improvements have been made for several iterations
of a different rule to break out of a cycle and ensure convergence.

One iteration of the simplex method’s optimization phase moves a vertex
partition to a neighboring vertex based on a heuristic for the entering index. Algo-
rithm 11.3 implements such an iteration with the greedy heuristic. Example 11.7
demonstrates using the simplex algorithm starting from a known vertex partition
to solve a linear program.

11.2.4 Initialization Phase
The optimization phase of the simplex algorithm is implemented in algorithm 11.4.
Unfortunately, algorithm 11.4 requires an initial partition that corresponds to a
vertex. If we do not have such a partition, we must solve an auxiliary linear program
to obtain this partition as part of an initialization phase.

The auxiliary linear program to be solved in the initialization phase includes
extra variables z ∈ Rm, which we seek to zero out:15 15 The values for z represent the

amount by which Ax = b is vi-
olated. By zeroing out z, we find
a feasible point. If, in solving the
auxiliary problem, we do not find
a vertex with a zeroed-out z, then
we can conclude that the prob-
lem is infeasible. Furthermore, it
is not always necessary to add
all m extra variables, especially
when slack variables are included
when transforming between stan-
dard and equality form.

minimize
x,z

[

0⊤ 1⊤
]
[

x

z

]

subject to
[

A Z
]
[

x

z

]

= b

[

x

z

]

≥ 0

(11.28)

Consider the equality-form linear program with

A =

[

1 1 1 0

−4 2 0 1

]

, b =

[

9

2

]

, c =








3

−1

0

0








and the initial vertex defined by B = {3, 4}. After verifying that B defines a
feasible vertex, we can begin one iteration of the simplex algorithm.

We extract xB :

xB = A−1
B b =

[

1 0

0 1

]−1 [

9

2

]

=

[

9

2

]

We then compute λ:

λ = A−⊤B cB =

[

1 0

0 1

]−⊤ [
0

0

]

= 0

and µV :

µV = cV −
(

A−1
B AV

)⊤
cB =

[

3

−1

]

−




[

1 0

0 1

]−1 [

1 1

−4 2

]



⊤ [
0

0

]

=

[

3

−1

]

Wefind that µV contains negative elements, so our current B is suboptimal.
We will pivot on the index of the only negative element, q = 2. An edge
transition is run from xB in the direction −A−1

B A{q} = [1, 2].
Using equation (11.19), we increase x′q until a new constraint becomes

active. In this case, x′q = 1 causes x4 to become zero. We update our set of
basic indices to B = {2, 3}.

In the second iteration, we find:

xB =

[

1

8

]

, λ =

[

0

−1/2

]

, µV =

[

1

1/2

]

.

The vertex is optimal because µV has no negative entries. Our algorithm
thus terminateswithB = {2, 3}, forwhich the design point is x∗ = [0, 1, 8, 0].

Example 11.7. Solving a linear pro-
gram with the simplex algorithm.

function step_lp!(B, LP)
A, b, c = LP.A, LP.b, LP.c
n = size(A, 2)
b_inds = sort!(B)
n_inds = sort!(setdiff(1:n, B))
AB, AV = A[:,b_inds], A[:,n_inds]
xB = AB\b
cB = c[b_inds]
λ = AB' \ cB
cV = c[n_inds]
μV = cV - AV'*λ

q, p, xq′, Δ = 0, 0, Inf, Inf
for i in 1 : length(μV)

if μV[i] < 0
pi, xi′ = edge_transition(LP, B, i)
if μV[i]*xi′ < Δ

q, p, xq′, Δ = i, pi, xi′, μV[i]*xi′
end

end
end
if q == 0

return (B, true) # optimal point found
end

if isinf(xq′)
error("unbounded")

end

j = findfirst(isequal(b_inds[p]), B)
B[j] = n_inds[q] # swap indices
return (B, false) # new vertex but not optimal

end

Algorithm 11.3. A single iteration
of the simplex algorithm in which
the set B is moved from one ver-
tex to a neighbor while maximally
decreasing the objective function.
Here, step_lp! takes a partition
defined by B and a linear program
LP.

function minimize_lp!(B, LP)
done = false
while !done

B, done = step_lp!(B, LP)
end
return B

end

Algorithm 11.4. Minimizing a lin-
ear program given a vertex parti-
tion defined by B and a linear pro-
gram LP.

where Z is a diagonal matrix whose diagonal entries are

Zii =







+1 if bi ≥ 0

−1 otherwise.
(11.29)

The auxiliary linear program is solved with a partition defined by B, which
selects only the z values. The corresponding vertex has x = 0, and each z-element
is the absolute value of the corresponding b-value: zj = |bj|. This initial vertex
can easily be shown to be feasible.

Example 11.8 demonstrates using an auxiliary linear program to obtain a
feasible vertex.

Consider the equality-form linear program:

minimize
x1,x2,x3

c1x1 + c2x2 + c3x3

subject to 2x1 − 1x2 + 2x3 = 1

5x1 + 1x2 − 3x3 = −2

x1, x2, x3 ≥ 0

We can identify a feasible vertex by solving:

minimize
x1,x2,x3,z1,z2

z1 + z2

subject to 2x1 − 1x2 + 2x3 + z1 = 1

5x1 + 1x2 − 3x3 − z2 = −2

x1, x2, x3, z1, z2 ≥ 0

with an initial vertex defined by B = {4, 5}.
The initial vertex has:

x
(1)
B = A−1

B bB =

[

1 0

0 −1

]−1 [

1

−2

]

=

[

1

2

]

and is thus x(1) = [0, 0, 0, 1, 2]. Solving the auxiliary problem yields x∗ ≈
[0.045, 1.713, 1.312, 0, 0]. Thus [0.045, 1.713, 1.312] is a feasible vertex in the
original problem.

Example 11.8. Using an auxiliary
linear program to obtain a feasible
vertex.

The partition obtained by solving the auxiliary linear program will produce a
feasible design point, Ax = b because z will have been zeroed out. If z is nonzero,
then the original linear program is infeasible. If z is zero, then the resulting
partition can be used as the initial partition for the optimization phase of the
simplex algorithm. The original problem must be slightly modified to incorporate
the new z variables:

minimize
x,z

[

c⊤ 0⊤
]
[

x

z

]

subject to

[

A I

0 I

] [

x

z

]

=

[

b

0

]

[

x

z

]

≥ 0

(11.30)

The z values must be included. Despite their vector counterparts being zero, it
is possible that some indices in the components of z are included in the initial
partition B. One can inspect the initial partition and include only the specific
components that are needed.

The solution (x∗, z∗) obtained by solving the second LP will have z∗ = 0. Thus,
x∗ will be a solution to the original linear problem.

An implementation for the complete simplex algorithm is given in algorithm 11.5.

11.3 Dual Certificates

Suppose we have a candidate solution and we want to verify that it is optimal.
Verifying optimality using dual certificates (algorithm 11.6) is useful in many cases,
such as debugging our linear program code.

We know from the FONCs for constrained optimization that the optimal value
of the dual problem d∗ is a lower bound of the optimal value of the primal problem
p∗. Linear programs are linear and convex, and one can show that the optimal
value of the dual problem is also the optimal value of the primal problem, d∗ = p∗.

function minimize_lp(LP)
A, b, c = LP.A, LP.b, LP.c
m, n = size(A)
z = ones(m)
Z = Matrix(Diagonal([j ≥ 0 ? 1 : -1 for j in b]))

A′ = hcat(A, Z)
b′ = b
c′ = vcat(zeros(n), z)
LP_init = LinearProgram(A′, b′, c′)
B = collect(1:m).+n
minimize_lp!(B, LP_init)

if any(i-> i > n, B)
error("infeasible")

end

A′′ = [A Matrix(1.0I, m, m);
zeros(m,n) Matrix(1.0I, m, m)]

b′′ = vcat(b, zeros(m))
c′′ = c′
LP_opt = LinearProgram(A′′, b′′, c′′)
minimize_lp!(B, LP_opt)
return get_vertex(B, LP_opt)[1:n]

end

Algorithm 11.5. The simplex algo-
rithm for solving linear programs
in equality form when an initial
partition is not known.

The primal linear program can be converted to its dual form:

max
µ≥0,λ

min
x
L(x,µ,λ) = max

µ≥0,λ
min

x
c⊤x− µ⊤x− λ⊤(Ax− b) (11.31)

= max
µ≥0,λ

min
x

(c− µ−A⊤λ)⊤x + λ⊤b (11.32)

From the FONC, we know c− µ−A⊤λ = 0, which allows us to drop the first
term in the objective above. In addition, we know µ = c − A⊤λ ≥ 0, which
implies A⊤λ ≤ c. In summary, we have:
Primal Form (equality) Dual Form

minimize
x

c⊤x

subject to Ax = b

x ≥ 0

maximize
λ

b⊤λ

subject to A⊤λ ≤ c

If the primal problem has n variables and m equality constraints, then the dual
problem has m variables and n constraints.16 Furthermore, the dual of the dual is 16 An alternative to the simplex al-

gorithm, the self-dual simplex algo-
rithm, tends to be faster in prac-
tice. It does not require that the ma-
trix AB satisfy xB = A−1

B b ≥ 0.
The self-dual simplex algorithm is
a modification of the simplex al-
gorithm for the dual of the linear
programming problem in standard
form.

the primal problem.
Optimality can be assessed by verifying three properties. If someone claims

(x∗,λ∗) is optimal, we can quickly verify the claim by checking whether all three
of the following conditions are satisfied:
1. x∗ is feasible in the primal problem.

2. λ∗ is feasible in the dual problem.

3. p∗ = c⊤x∗ = b⊤λ∗ = d∗.
Dual certificates are used in example 11.9 to verify the solution to a linear

program.

function dual_certificate(LP, x, λ, ϵ=1e-6)
A, b, c = LP.A, LP.b, LP.c
primal_feasible = all(x .≥ 0) && A*x ≈ b
dual_feasible = all(A'*λ .≤ c)
return primal_feasible && dual_feasible &&

isapprox(c⋅x, b⋅λ, atol=ϵ)
end

Algorithm 11.6. A method for
checking whether a candidate so-
lution given by design point x and
dual point λ for the linear program
LP in equality form is optimal. The
parameter ϵ controls the tolerance
for the equality constraint.

Consider the standard-form linear program with

A =






1 1 −1

−1 2 0

1 2 3




 , b =






1

−2

5




 , c =






1

1

−1






We would like to determine whether x∗ = [2, 0, 1] and λ∗ = [1, 0, 0] are an
optimal solution pair. We first verify that x∗ is feasible:

Ax∗ = [1,−2, 5] = b, x∗ ≥ 0

We then verify that λ∗ is dual-feasible:

A⊤λ∗ = [1, 1,−1] ≤ c

Finally, we verify that p∗ and d∗ are the same:

p∗ = c⊤x∗ = 1 = b⊤λ∗ = d∗

We conclude that (x∗,λ∗) are optimal.

Example 11.9. Verifying a solution
using dual certificates.

11.4 Summary

• Linear programs are problems consisting of a linear objective function and
linear constraints.

• The simplex algorithm can optimize linear programs globally in an efficient
manner.

• Dual certificates allow us to verify that a candidate primal-dual solution pair
is optimal.

11.5 Exercises

Exercise 11.1. Suppose you do not know any optimization algorithm for solving
a linear program. You decide to evaluate all the vertices and determine, by in-
spection, which one minimizes the objective function. Give a loose upper bound
on the number of possible minimizers you will examine. Furthermore, does this
method properly handle all linear constrained optimization problems?

Exercise 11.2. If the program in example 11.1 is bounded below, argue that the
simplex method must converge.

Exercise 11.3. Suppose we want to minimize 6x1 + 5x2 subject to the constraint
3x1 − 2x2 ≥ 5. How would we translate this problem into a linear program in
equality form with the same minimizer?

Exercise 11.4. Suppose your optimization algorithm has found a search direction
d and you want to conduct a line search. However, you know that there is a linear
constraint w⊤x ≥ 0. Howwould youmodify the line search to take this constraint
into account? You can assume that your current design point is feasible.

Exercise 11.5. Reformulate the linear program

minimize
x

c⊤x

subject to Ax ≥ 0
(11.33)

into an unconstrained optimization problem with a log barrier penalty.

